
    Cairnloop 🏔️

[image: Hex.pm Version]
[image: HexDocs]
[image: GitHub Actions CI]
An embedded, Phoenix-native customer support automation layer for Elixir applications.
Installation
The fastest way to install Cairnloop is with the Igniter installer. First, add Igniter to your
dependencies if it is not already present, then run:
mix deps.get
mix cairnloop.install

The installer adds {:cairnloop, "~> 0.1.0"} to your mix.exs deps and generates a
create_cairnloop_tables migration against your detected Ecto repo.
Manual install (without Igniter)
Add Cairnloop to your mix.exs dependencies:
def deps do
  [
    {:cairnloop, "~> 0.1.0"}
  ]
end
Then run mix deps.get and generate the migration manually (see the quickstart guide for the
migration contents).
Why Cairnloop?
	Host-owned support truth. Conversations, drafts, governed actions, and outbound follow-ups
all live in your Postgres database — no external CRM sync required.
	Safe automation by default. The AI drafts; your operators approve. Every proposed action
passes through a durable approval state machine before execution.
	KB substrate built in. Hybrid pgvector + full-text retrieval keeps AI answers grounded in
your published knowledge base, not hallucinated context.
	Additive, not invasive. Cairnloop is a library embedded in your existing Phoenix monolith.
No separate service, no separate deploy.
	Observable without lock-in. Bounded :telemetry spans on every meaningful operation.
Export to any APM; your app stays in control.

What it does
Cairnloop turns support conversations into answers, product signals, knowledge-base improvements,
and safe automated actions — all inside your existing Phoenix app. Incoming messages route through
Phoenix Channels to a durable Ecto-backed conversation record. An Oban worker drafts an
AI-grounded reply using your published KB and offers it to the operator for approval. Resolved
conversations feed back into the KB maintenance queue through a governed review workflow, and
support-triggered outbound follow-ups route durably through a configurable Notifier behaviour.
Explore the guides
	Quickstart — Clone, install, boot the
example app, and walk your first support conversation end-to-end.
	JTBD Walkthrough — A prose walkthrough
of every Jobs-To-Be-Done stage in the seeded example: inbox → draft approval → governed tool
proposal → resolve → outbound follow-up → bulk recovery.
	Host Integration — Wire up
ContextProvider, Notifier, AutomationPolicy, and SLAPolicyProvider in your app. Includes
telemetry patterns and Oban configuration.
	Troubleshooting — Common install errors,
migration order, pgvector setup, and Oban worker timing.

Contributing
Contributions are welcome. Open an issue or pull request on
GitHub. Please follow the existing code style and run
mix test before submitting.
License
MIT. See LICENSE.


  

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
0.5.1 (2026-06-03)
Bug Fixes
	inject the signed-in operator as host_user_id, not a hardcoded id (dce852a)
	inject the signed-in operator as host_user_id, not a hardcoded id (72891f5)

0.5.0 (2026-06-03)
Features
	automation: swappable draft-generator seam + Anthropic adapter (6632b4e)
	automation: swappable draft-generator seam + Anthropic adapter (ea9bf96)

0.4.0 (2026-06-03)
Features
	ui: add Cockpit Home + persistent nav shell (IA orientation layer) (fa24f9d)
	ui: design-system + IA elevation for the operator UI (ebed003)
	ui: finish Conversation Workspace rail on the design system (Pass 6) (d680ca4)
	ui: rebuild Audit Log on the design system (Pass 3, screen 2) (517da98)
	ui: rebuild Conversation Workspace on the design system (Pass 3) (8b4c362)
	ui: rebuild Settings on the design system (Pass 3, screen 1) (71ad202)
	ui: rebuild the Knowledge Base cluster on the design system (Pass 3) (f543de4)
	ui: rebuild the operator Inbox on the design system (Pass 3) (e40dc55)
	ui: ship cairnloop.css design system + shared component library (0c96516)

Bug Fixes
	ui: green the integration gate — restore bare brand tokens (eb73bb2)

0.3.0 (2026-05-30)
Features
	add mix cairnloop.doctor wiring diagnostic (24e71f9)
	installer: print router/auditor/doctor next steps after install (bcacf69)
	mix cairnloop.doctor + validated router opts + installer next-steps (Tier 2 DX) (7f559c2)
	router: validated, self-documenting macro options + live_session name (5524e82)

0.2.3 (2026-05-30)
Bug Fixes
	router: cairnloop_dashboard/2 compile break + verify-before-publish hardening (21699d5)
	router: cairnloop_dashboard/2 failed to compile for adopters (dc0784f)

0.2.2 (2026-05-30)
Bug Fixes
	governance: preserve approver as decided_by through execute co-commit (607eaa5)
	integration: green the DB-backed integration CI suite (9d4fedd)
	web: humanize audit-log action-filter options — no raw atom leak (a6eb451)

0.2.1 (2026-05-30)
Bug Fixes
	0.2.1: repair AUDIT-01 audit log + OPS endpoints; adopt release-please (bb74b1a)
	0.2.1: repair AUDIT-01 audit log + OPS health/metrics mounting (62b7989)

Unreleased
0.2.0 - 2026-05-29
Added
	Security domain closure: Cairnloop.KnowledgeAutomation strictly rejects spoofed,
already-published, and caller-supplied-grounding inputs — suggestions reuse only non-published
targets, gap-candidate grounding derives exclusively from hydrated evidence, and stale-gate
inputs load only from repo-backed GapEvent rows (Phase 33, SEC-01/02/03)
	Operator Settings Surface: SettingsLive gains real MCP token management (CRUD, masking,
validation), Notifier and retrieval health indicators, and a persisted dark-mode toggle
(Phase 34, SET-01/02/03/04)
	Audit & operations support: operator Audit Log view and governed-actions rail pagination
(Phase 35, AUDIT-01, TECH-01)
	HTTP /health and /metrics endpoints via Cairnloop.Web.HealthPlug and
Cairnloop.Web.MetricsPlug (Phase 35, OPS-01, OPS-02)
	Documentation: guides/05-mcp-clients.md, guides/06-extending.md, root CONTRIBUTING.md,
and docs/architecture.md (Phase 36, DOC-01/02/03/04)
	Realistic demo fixtures: 12–16 seeded conversations spanning all JTBD states, 5 KB articles with revisions, 3 GapCandidates, 1 ArticleSuggestion ready for review (Phase 27)
	Customer /chat widget wired to real ingress via Cairnloop.Channels.WidgetSocket + WidgetChannel; two-tab demo (Phase 28)
	Brand-token CSS extraction: prompts/cairnloop.css :root block in example app; var(--cl-token) without hex fallback; negative-grep gate (Phase 29, D-10 closure)
	KB editorial polish: shared nav shell across 4 KB routes, "Create new article" affordance, gap-evidence sidebar in Editor, calm copy on SuggestionReview handoff (Phase 30)
	T-10-09 and T-10-11 closure: EditorHandoff double-layer gate (DB manual_edit_opened_at timestamp + signed token assertion) prevents preloading proposed_markdown without deliberate handoff (Phase 30)
	Golden-path JTBD smoke test in CI: golden_path_test.exs (E2E-01) + widget_channel_test.exs (E2E-02) under mix test.integration (Phase 31)
	README rewritten as an Igniter-first front door; four ExDoc guides (quickstart, JTBD walkthrough, host integration, troubleshooting) published to HexDocs (Phase 32)

0.1.0 - 2026-05-25
Added
	Host-owned hybrid retrieval corpus (pgvector + PG full-text) via Cairnloop.Retrieval
	Operator search with trust, recency, and citation cues
	Citation-backed grounded drafting with clarification and escalation states
	Durable gap-event storage and ranked KB gaps dashboard
	AI-prepared KB draft/revision suggestions with stale-revision gating and citation validation
	Review-gated KB update workflow: approve, reject, defer, publish — with append-only task event history
	In-thread quick-fix KB maintenance launched from live support conversations
	Host-owned governed-action contract: compile-time use Cairnloop.Tool with risk tiers and deny-by-default authorize/2
	Durable ToolProposal + ToolActionEvent records with Stripe-style idempotency
	Approval state machine with Oban-backed resume, expiry, and deferral paths
	Three-layer at-most-once execution: Oban unique + terminal guard + SHA-256 per-attempt run key
	Bounded [:cairnloop, :retrieval, …] and Cairnloop.Governance.Telemetry event namespaces
	Read-only MCP seam (tools/list, initialize) via optional Cairnloop.Web.MCP.Router Plug



  

    Quickstart

Get from a fresh clone to a running Cairnloop operator dashboard in a few minutes.
This guide follows the example app at examples/cairnloop_example/ as its reference.
Prerequisites
	Elixir 1.15+ / OTP 26+
	Postgres 16+ with the pgvector extension installed

The pgvector extension powers Cairnloop's Knowledge Base embeddings. If you need a
containerized Postgres with pgvector, the repository ships a docker-compose.yml at the
repo root. From the repo root run:
docker compose up -d db

Install
Cairnloop ships an Igniter installer that adds the
dependency and generates the database migration for you.
First, fetch deps:
mix deps.get

Then run the installer:
mix cairnloop.install

The installer does two things:
	Adds {:cairnloop, "~> 0.1.0"} to your mix.exs dependencies.
	Detects your Ecto repo via Igniter.Libs.Ecto.select_repo/1 and generates a
create_cairnloop_tables migration that creates cairnloop_conversations and
cairnloop_messages with the correct schema.

If no Ecto repo is found, the installer emits:
No Ecto repo found. Please create a migration manually for cairnloop tables.
In that case, create a priv/repo/migrations/<timestamp>_create_cairnloop_tables.exs
migration by hand before running mix ecto.migrate.
Run migrations
After the installer has run, apply both your host app's migrations and the Cairnloop
library's own migrations:
# Run host migrations (generated by the installer or written by hand)
mix ecto.migrate

# Run the Cairnloop library's own migrations
mix ecto.migrate --migrations-path deps/cairnloop/priv/repo/migrations

The library ships 15+ additional migrations (knowledge base, retrieval corpus, gap
candidates, article suggestions, outbound, and more) that mix ecto.migrate alone will
not apply. Skipping this second command will cause Postgrex.Error relation-not-found
errors the first time any non-chat feature is exercised.
Tip: Add both commands to your ecto.setup alias in mix.exs so they always run
together:
"ecto.setup": ["ecto.create", "ecto.migrate",
  "ecto.migrate --migrations-path deps/cairnloop/priv/repo/migrations",
  "run priv/repo/seeds.exs"]

Manual install (without Igniter)
If you prefer not to use Igniter, add Cairnloop to your deps directly:
# mix.exs
def deps do
  [
    {:cairnloop, "~> 0.1.0"}
  ]
end
Then create the create_cairnloop_tables migration manually.
Mount the Dashboard
Import the cairnloop_dashboard/2 macro and mount it under a scope in your router.
The example app uses /support:
# lib/my_app_web/router.ex
import Cairnloop.Router, only: [cairnloop_dashboard: 2]

scope "/support", MyAppWeb do
  pipe_through :browser

  cairnloop_dashboard("/", session: %{"host_user_id" => "demo_operator"})
end
The cairnloop_dashboard/2 macro mounts routes under whatever path you pass as its
first argument. With the /support scope above:
	Cockpit Home (the task-oriented landing) is at /support
	Inbox is at /support/inbox
	A conversation is at /support/:id
	Knowledge Base is at /support/knowledge-base
	Settings is at /support/settings

This guide assumes /support as in the example app. Adjust paths to match your own
scope if you mount elsewhere.
Route convention: Always use the path you pass to cairnloop_dashboard/2.
The internal integration-test routes (not the shipped macro routes) will 404 for adopters.

Styling
Cairnloop ships a self-contained, themeable stylesheet in the package at
priv/static/cairnloop.css. It defines the design tokens (--cl-*) and the .cl-*
component classes the dashboard renders — no Tailwind or daisyUI required in your app.
Include it once.
If you bundle CSS (esbuild/Tailwind), import it from your own stylesheet:
@import "../../deps/cairnloop/priv/static/cairnloop.css";
Or serve the packaged asset directly and link it from your root layout:
# endpoint.ex
plug Plug.Static, at: "/cairnloop", from: {:cairnloop, "priv/static"}, only: ["cairnloop.css"]
<link rel="stylesheet" href="/cairnloop/cairnloop.css" />
Theming: override any --cl-* token in CSS loaded after cairnloop.css to re-skin
the dashboard to your brand, and toggle dark mode by setting data-theme="dark" on <html>.
Boot
The commands below are for the example app. Switch into it first, then set up the
database and start the server:
cd examples/cairnloop_example
mix setup
mix phx.server

Then visit http://localhost:4000 for the guided demo index — it frames
the Trailmark scenario and links to every stage of the JTBD Walkthrough.
Or jump straight to the operator inbox at
http://localhost:4000/support.
You should see the Cairnloop operator inbox. The example app's seed places 20 conversations across
all lifecycle states — including ones pre-positioned in each JTBD state — so every screen is live
and clickable immediately.
Port already in use? Start the server with a different port: PORT=4010 mix phx.server
(and set PGPORT if your Postgres isn't on 5433). The example honors both.

Next Steps
	JTBD Walkthrough — walk the full Jobs-To-Be-Done
lifecycle in the seeded example: inbox → conversation workspace → cmd+k search →
AI draft approval → governed tool approval → resolve → outbound trigger → bulk recovery.
	Host Integration — implement the four host behaviour
contracts (ContextProvider, Notifier, AutomationPolicy, SLAPolicyProvider)
so Cairnloop knows your app's context and policy.
	Auth & Operator Identity — wire your real
authenticated operator into the dashboard so the audit log attributes actions correctly.
Read this before going past the demo — the hardcoded host_user_id shown in quickstarts
is a trap in production.
	Troubleshooting — resolve common install, migration,
pgvector, and mount-config errors.



  

    JTBD Walkthrough

This guide walks through the full Jobs-To-Be-Done (JTBD) lifecycle that Cairnloop supports
out of the box. You can follow every stage interactively by running the example app
(cd examples/cairnloop_example && mix setup && mix phx.server) and navigating alongside.
Route convention: All support routes are mounted under the path you pass to
cairnloop_dashboard/2 in your router. This guide assumes /support — the example app's
default. The customer chat widget lives at /chat.

The example app opens on a guided demo index that frames the scenario and links to every stage
below — a good place to start clicking around:
[image: Cairnloop demo index — the Trailmark scenario and a guided tour of all nine JTBD stages]
The screenshots in this guide are captured from the seeded example app. To refresh them, see
examples/cairnloop_example/screenshots/.


Stage 1: Seed — a customer message arrives
A customer submits a message through the embedded chat widget at /chat. On the customer side,
the Phoenix Channel (Cairnloop.Channels.WidgetSocket) receives the payload, persists a new
Conversation row with status :open, and appends an initial Message record with role :user.
No operator action is needed at this point. Cairnloop has recorded the inbound event durably in
Ecto. From here, the conversation is visible across any operator session.
[image: The customer-facing chat widget at /chat]

Stage 2: Inbox sees the conversation
The operator navigates to /support. The inbox rendered by Cairnloop.Web.InboxLive loads all
conversations the current operator can see.
Screen region: The inbox lists the seeded conversations spanning all status states — new,
open, awaiting customer, and resolved. Each row shows the conversation subject, the customer
identifier, the time since the last message, and a status chip. The just-seeded conversation
appears at or near the top of the list.
[image: The operator inbox at /support, listing conversations across the lifecycle]
At this stage, the operator can orient themselves across the support queue before selecting a
conversation to work.

Stage 3: Conversation workspace — cmd+k search and citation chip
The operator clicks a conversation row and is taken to /support/:id — the conversation
workspace rendered by Cairnloop.Web.ConversationLive.
Screen region: The workspace splits into a main column (the conversation timeline: customer
messages, operator replies, AI draft cards, action event cards, and outbound bubbles) and a right
rail (customer context from ContextProvider, SLA status, governed-action history).
[image: The conversation workspace: timeline plus the customer-context rail]
The operator presses cmd+k to open the search palette. They type a query — for example,
"refund" — and Cairnloop retrieves matching Knowledge Base articles and resolved support
evidence via the retrieval backend. Results appear with source, recency, and trust cues.
Screen region: The cmd+k modal shows a list of retrieval results. Each result card displays
the article title, a short content excerpt, the source type (knowledge base or resolved case),
and a trust level chip. Clicking a result activates it and places a citation chip in the
operator's draft compose area.
The citation chip links the operator's eventual reply back to a grounded retrieval source —
meaning no freeform AI output leaves the thread without an auditable evidence anchor.

Stage 4: Approve AI draft
Cairnloop's AI drafting engine prepares a suggested reply for the conversation. The draft
appears in the conversation timeline as a Draft card in :pending status, surfaced via
Cairnloop.Automation.
Screen region: The AI draft card shows the proposed reply text, the draft source
(knowledge base articles or prior resolved cases that grounded the reply), and two actions:
"Approve" and "Dismiss". An approval badge shows that a human decision is required before
anything is sent to the customer.
[image: A pending AI draft awaiting operator approval, with its grounding and customer context]
The operator reviews the draft text and evidence, then clicks "Approve". The draft status
transitions to :approved. The approved reply is appended to the conversation timeline and
becomes part of the durable support record.
No AI reply is sent to the customer without explicit operator approval. This is the
human-in-the-loop (HITL) guarantee at the reply layer.

Stage 5: Governed tool proposal approve
The operator encounters a situation requiring a structured action — for example, updating an
account field or triggering a downstream system event. Cairnloop's governed-tool contract
routes this through Cairnloop.Governance.propose/3, which records a durable ToolProposal
row with a snapshotted risk tier, approval mode, and input before any execution occurs.
Screen region: The conversation timeline shows a governed-action card in :pending_approval
state. The card displays the tool title, a risk tier chip (e.g., "Low Write"), the proposed
inputs, and an approve/reject affordance. Trust facts are snapshotted at proposal time and do
not change between now and execution.
[image: A governed action proposed and waiting in the approval lane]
The operator reviews the proposal and clicks "Approve". The approval state machine in
Cairnloop.Governance records the decision via Governance.approve/3, transitions the
ToolApproval record, and enqueues the next step via Oban.
No governed action executes inline. Every proposal waits for explicit operator sign-off before
the execution worker runs.

Stage 6: ToolExecutionWorker reaches :success
With approval granted, Cairnloop.Workers.ToolExecutionWorker runs the approved action. The
worker applies three-layer at-most-once idempotency: Oban unique keys, a terminal guard on the
worker itself, and a SHA-256 per-attempt run key.
Screen region: The governed-action card in the conversation timeline updates to show
"Action completed" with an execution timestamp. The risk tier chip changes to a success state.
A full append-only ToolActionEvent history is recorded on the proposal.
[image: The approved action after execution — &quot;Action completed&quot;]
The execution result is part of the durable conversation record. If the worker were to fail or
be retried, the idempotency guarantees prevent the action from running more than once.

Stage 7: Resolve the conversation
The operator closes out the support thread by resolving the conversation. Cairnloop calls
Cairnloop.Chat.resolve_conversation/2 directly (resolution is a domain function, not a
LiveView event), which transitions the Conversation status to :resolved and records the
resolution timestamp and actor.
Screen region: The conversation status chip in the workspace header changes to "Resolved".
The conversation becomes ineligible for further AI drafting. A resolution timestamp and operator
identifier appear in the thread metadata rail.
[image: A resolved conversation carrying its outcome and CSAT signal]
At resolution, Cairnloop emits both a bounded :telemetry span and a [:cairnloop, :conversation, :resolved]
domain event. The host Notifier behaviour receives on_conversation_resolved/2 — see
Host Integration for the full callback set.

Stage 8: Outbound trigger from the sidebar
Now that the conversation is resolved, the operator notices a recovery follow-up opportunity
in the sidebar. The sidebar renders a "Send Recovery Follow-up" button for resolved
conversations that have not yet received a follow-up outbound message.
Screen region: The right rail of the resolved conversation shows a "Send Recovery
Follow-up" button. Clicking it opens a confirmation affordance that previews the outbound
message body (rendered from the configured outbound_recovery_template_id).
The operator confirms and Cairnloop calls Cairnloop.Outbound.trigger/2, which:
	Appends a system_outbound message to the conversation timeline (with role: :system_outbound).
	Enqueues an OutboundWorker Oban job that routes through the host's Notifier.on_outbound_triggered/2.

Screen region: A new outbound bubble appears in the conversation timeline with a "Pending"
delivery chip. After the Oban worker runs, the chip transitions to "Sent" or "Failed"
depending on the Notifier return value.
[image: A resolved conversation with a durable outbound recovery follow-up pending delivery]

Stage 9: Bulk recovery
The operator navigates back to /support and realizes there are several more resolved
conversations that need follow-up. Cairnloop supports bulk outbound via InboxLive
multi-select.
Screen region: In the inbox, each row has a checkbox. The operator selects multiple
resolved conversations. A bulk action bar appears at the bottom of the inbox listing the
selected count and a "Send recovery follow-up" button.
[image: Bulk recovery: multiple resolved conversations selected with the bulk action bar]
The operator clicks "Send Recovery Follow-up". A confirmation modal appears:
Screen region: The modal shows a preview of the outbound message body, a sample of the
first five recipient identifiers, and a count of the total cohort size. If the cohort exceeds
Cairnloop.Outbound.max_batch_size (25), a fail-closed refusal banner replaces the confirm
button — the bulk send is refused rather than silently capping the batch.
The operator confirms. Cairnloop calls Cairnloop.Outbound.bulk_trigger/2, which:
	Records a BulkEnvelope audit row with status :submitted (or :refused_cap_exceeded if
the cap was exceeded).
	Enqueues one OutboundWorker Oban job per recipient, each with an at-most-once Oban unique
key scoped to (conversation_id, template_id, bulk_envelope_id).

Screen region: The bulk action bar disappears and the inbox deselects all rows. Each
recipient conversation now has a pending outbound bubble in its timeline. Operators can inspect
the BulkEnvelope row via the Governance audit facade
(Cairnloop.Governance.list_recent_bulk_outbound_envelopes/1).

This walkthrough covers all nine stages of the JTBD lifecycle from first customer message
through bulk recovery fan-out. The same sequence is exercised in CI by the integration smoke
test at test/integration/golden_path_test.exs.

Beyond the thread: the supporting surfaces
The same operator dashboard exposes the surfaces that turn day-to-day support into durable
knowledge and an auditable trail.
Knowledge base (/support/knowledge-base) — published articles and their revision history:
[image: The knowledge-base editorial surface]
Knowledge gaps (/support/knowledge-base/gaps) — recurring unanswered patterns surfaced from
real conversations, ready to become new articles:
[image: The knowledge-gaps queue]
Audit log (/support/audit-log) — the append-only timeline of every governed-action event,
newest first:
[image: The governed-action audit log]
Settings (/support/settings) — MCP tokens, Notifier health, the retrieval index, and Oban
job state:
[image: The operator settings cockpit]
Refreshing these screenshots: boot the seeded example app
(cd examples/cairnloop_example && mix ecto.reset && mix phx.server) and run the capture tool in
examples/cairnloop_example/screenshots/ (npm install && npm run capture). It drives the demo
with Playwright and rewrites guides/assets/. The capture is non-gating and asserts nothing —
the deterministic test/integration/golden_path_test.exs remains the source of CI truth.



  

    Host Integration

Cairnloop exposes four behaviour contracts that let your host application control the
support lifecycle without giving up ownership of your data or business logic. Each behaviour
is a plain Elixir @behaviour module — implement the callbacks, configure the module, and
Cairnloop uses your implementation at the right moment.
This guide documents all four behaviours and their full callback sets, in the order a new
adopter would typically implement them. It also covers Cairnloop's telemetry emission points
as an observability reference.

ContextProvider
Module: Cairnloop.ContextProvider
When to implement: Always — every operator conversation workspace pulls context from this
behaviour. Without a configured provider, the context rail renders "Context Unavailable".
ContextProvider is how Cairnloop achieves a zero-API-sync design. Your implementation
returns a map of host-owned facts for the given support actor. Cairnloop renders that map
recursively as categorized UI sections in the conversation workspace's right rail — no
frontend code required on your end. This is the "Zero-Config UI" feature.
Callback:
@callback get_context(actor_id :: String.t(), opts :: keyword()) ::
            {:ok, map()} | {:error, term()}
The actor_id is the raw string from the Cairnloop conversation. Your implementation maps
it to your internal domain — a UUID, integer ID, email, or external identifier. The opts
keyword list is reserved for future extension and can be ignored.
Return a tagged tuple. Never raise. If your database or external service is unavailable,
return {:error, reason} — the dashboard degrades to "Context Unavailable" rather than
crashing the operator's session.
Example implementation:
defmodule MyApp.CairnloopContext do
  @behaviour Cairnloop.ContextProvider

  @impl true
  def get_context(actor_id, _opts) do
    case MyApp.Accounts.get_user(actor_id) do
      nil ->
        {:ok, %{}}

      user ->
        {:ok, %{
          "User Details" => %{name: user.name, lifetime_value: "$#{user.ltv}"},
          "Active Plan"  => %{tier: user.plan, status: user.billing_status}
        }}
    end
  end
end
Configuration (config/config.exs):
config :cairnloop, :context_provider, MyApp.CairnloopContext
The returned map is rendered as grouped sections in the support workspace right rail. Top-level
string keys become section headers; nested maps render as key-value pairs within that section.

Notifier
Module: Cairnloop.Notifier
When to implement: Required when you need side effects on conversation events, and required
for the outbound delivery lane (Outbound.trigger/2 and bulk_trigger/2 route through
on_outbound_triggered/2).
Notifier is the business-logic integration point: CRM sync, email, webhooks, background
jobs, or any other side effect your app needs to trigger when support events occur. Cairnloop
calls these callbacks from within Oban workers — on_conversation_resolved/2 and
on_outbound_triggered/2 each have a dedicated Oban worker, while on_sla_breach/3 is
called directly within the SLA countdown worker. Raising in any callback retries the
enclosing Oban job.
Callbacks (all three — implement all of them):
@callback on_conversation_resolved(conversation :: struct(), metadata :: map()) ::
            :ok | any()

@callback on_sla_breach(conversation :: struct(), sla :: struct(), metadata :: map()) ::
            :ok | {:error, term()} | any()

@callback on_outbound_triggered(message :: struct(), conversation :: struct()) ::
            :ok | {:error, term()} | any()
Example implementation:
defmodule MyApp.CairnloopNotifier do
  @behaviour Cairnloop.Notifier

  @impl true
  def on_conversation_resolved(conversation, metadata) do
    actor = metadata[:actor]

    # Enqueue a background job rather than performing the side effect inline.
    %{conversation_id: conversation.id, resolved_by_id: actor && actor.id}
    |> MyApp.Workers.CRMSyncJob.new()
    |> Oban.insert()

    :ok
  end

  @impl true
  def on_sla_breach(_conversation, _sla, _metadata) do
    # Notify your on-call channel, update a dashboard, or page an operator.
    :ok
  end

  @impl true
  def on_outbound_triggered(_message, _conversation) do
    # Route the outbound message through your delivery provider (email, SMS, etc.).
    # Return :ok on success or {:error, reason} to signal a delivery failure to Oban.
    :ok
  end
end
Generator escape hatch: Rather than writing this module by hand, run:
mix cairnloop.gen.notifier

This scaffolds a MyApp.CairnloopNotifier module with all three callbacks and automatically
injects the configuration line into your config/config.exs.
Manual configuration (config/config.exs):
config :cairnloop, :notifier, MyApp.CairnloopNotifier
The README's earlier examples showed only two Notifier callbacks. The full behaviour
defines three: on_conversation_resolved/2, on_sla_breach/3, and
on_outbound_triggered/2. Implement all three — missing callbacks cause a compile-time
warning about incomplete behaviour implementation.


AutomationPolicy
Module: Cairnloop.AutomationPolicy
When to implement: When you want to control how Cairnloop handles AI draft proposals.
Without a configured policy, the default posture is :require_approval — every AI draft
waits for explicit operator review before anything is sent.
AutomationPolicy is the governance boundary for AI drafting. Your implementation receives
a proposal map and opts, and returns one of four atoms that determine what Cairnloop does
with the AI-generated content.
Callback:
@callback decide(proposal :: map(), opts :: map()) ::
            :allow | :draft_only | :require_approval | :deny
Return values:
	Atom	Meaning
	:allow	The draft may be sent without operator review (use with caution — bypasses HITL).
	:draft_only	Cairnloop prepares the draft but does not surface an approval prompt. Operator must retrieve and act on it manually.
	:require_approval	The draft requires explicit operator approval before it becomes an outgoing reply. This is the recommended default.
	:deny	Cairnloop discards the draft. No AI content is presented to the operator for this proposal.

Example implementation (recommended — approval-gated):
defmodule MyApp.CairnloopPolicy do
  @behaviour Cairnloop.AutomationPolicy

  @impl true
  def decide(_proposal, _opts), do: :require_approval
  # Returns :allow | :draft_only | :require_approval | :deny
end
Start with :require_approval. This mirrors Cairnloop's "safe-by-default, not
autonomous-by-default" posture — operators review before anything reaches a customer.
Graduate to finer-grained logic (inspecting proposal.risk_tier, conversation tags, or
account properties) only after you have validated quality in your specific context.
Configuration (config/config.exs):
config :cairnloop, :automation_policy, MyApp.CairnloopPolicy

SLAPolicyProvider
Module: Cairnloop.SLAPolicyProvider
When to implement: When your support team has SLA commitments — response time targets,
breach thresholds, or priority tiers — that you want Cairnloop to track and enforce.
SLAPolicyProvider supplies Cairnloop with the active SLA rule set at runtime. Policies can
be stored in your database and retrieved dynamically, letting you change SLA terms without
redeploying.
Callbacks:
@callback get_active_policies() :: {:ok, list(map())} | {:error, term()}

@callback set_policy(priority :: atom(), attrs :: map()) :: {:ok, map()} | {:error, term()}
Example implementation:
defmodule MyApp.CairnloopSLA do
  @behaviour Cairnloop.SLAPolicyProvider

  @impl true
  def get_active_policies do
    # Return a list of SLA policy maps from your database or config.
    # Return {:ok, []} to disable SLA tracking.
    {:ok, [
      %{priority: :high, response_minutes: 60, breach_minutes: 240},
      %{priority: :normal, response_minutes: 240, breach_minutes: 1440}
    ]}
  end

  @impl true
  def set_policy(_priority, _attrs) do
    # Persist a new or updated SLA policy. Return {:ok, policy_map} on success.
    {:error, :not_implemented}
  end
end
Configuration (config/config.exs):
config :cairnloop, :sla_policy_provider, MyApp.CairnloopSLA
When get_active_policies/0 returns {:ok, []}, Cairnloop disables SLA breach tracking
rather than raising — the support workflow continues without SLA enforcement.

Operations endpoints (health & metrics)
Cairnloop ships two plain plugs for infrastructure monitoring, mounted with one router
helper:
	GET /health → Cairnloop.Web.HealthPlug — liveness/readiness probe; returns
200 with {"status": "ok"}.
	GET /metrics → Cairnloop.Web.MetricsPlug — Prometheus text exposition. Returns the
scrape when :telemetry_metrics_prometheus_core is running, or 501 until you add and
start it.

Mounting (MyAppWeb.Router):
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  require Cairnloop.Router

  # Mount outside your auth pipeline so infrastructure can reach the probes.
  scope "/" do
    Cairnloop.Router.cairnloop_operations()
  end
end
Override the paths if /health or /metrics collide with your own routes:
Cairnloop.Router.cairnloop_operations(health_path: "/healthz", metrics_path: "/internal/metrics")
Enabling /metrics: add the optional dependency and start the reporter in your
supervision tree so the plug has metrics to scrape:
# mix.exs
{:telemetry_metrics_prometheus_core, "~> 1.2"}
# application.ex children
{TelemetryMetricsPrometheus.Core, metrics: MyApp.Telemetry.metrics()}
Without the reporter, /health still works and /metrics returns 501 with guidance
rather than crashing.

Telemetry (observability only)
Cairnloop emits :telemetry events for observability. Per the project's architecture
posture: telemetry is observability only — never a UI or display source. Do not build
logic that reads Cairnloop telemetry to drive LiveView state; use the Notifier behaviour
for business-logic side effects.
Cairnloop uses a dual emission architecture: bounded :telemetry.span/3 events for
APM tracing alongside past-tense domain events for business logic hooks.
A. Tracing spans (performance and APMs)
Use the span lifecycle events (:start, :stop, :exception) to capture execution
metrics. These are appropriate for exporting to APMs (Datadog, Prometheus) or logging
function execution duration.
:telemetry.attach(
  "cairnloop-apm-tracker",
  [:cairnloop, :conversation, :resolve, :stop],
  fn _event, measurements, _metadata, _config ->
    require Logger
    # Execution time is available in measurements.duration (native units).
    Logger.info(
      "Resolve took #{System.convert_time_unit(measurements.duration, :native, :millisecond)}ms"
    )
  end,
  nil
)
B. Domain events (business lifecycle hooks)
Use past-tense domain events to observe successful business actions. These events carry
the resolved Conversation struct and actor metadata.
:telemetry.attach(
  "cairnloop-domain-hooks",
  [:cairnloop, :conversation, :resolved],
  fn _event, measurements, metadata, _config ->
    require Logger

    conversation = metadata.conversation
    Logger.info(
      "Conversation #{conversation.id} resolved by #{metadata.actor.id} " <>
      "in #{measurements.duration_seconds}s at #{conversation.resolved_at}"
    )

    # Example: broadcast to a LiveView session to surface a CSAT prompt.
    # Phoenix.PubSub.broadcast(
    #   MyApp.PubSub,
    #   "user_sessions:#{metadata.host_user_id}",
    #   :support_issue_resolved
    # )
  end,
  nil
)
Telemetry event names follow the [:cairnloop, :domain, :action, :lifecycle] convention.
The events are non-blocking — they do not delay the conversation resolve path. For
side-effects that need reliability guarantees (retries, durability), use Notifier
instead.

